

# TIMER/COUNTER

---

# Timer/Counter

---

- 8051 có 2 Timer
  - Timer 0
  - Timer 1
- Mỗi Timer là một bộ đếm 16 bit
  - Phạm vi đếm: từ 0000h đến FFFFh
  - Cờ tràn (TF) được thiết lập bằng 1 khi đếm từ FFFFh về 0000H

# Timer/Counter

---

- Có 3 chức năng:
  - Sử dụng để tạo thời gian **trễ** (khoảng thời gian giữa hai sự kiện): dùng xung Clock bên trong
  - **Đếm sự kiện**: sử dụng nguồn xung từ bên ngoài
    - Ví dụ:
      - Đếm số lượng người qua cửa
      - Đếm số vòng quay của động cơ
      - Hay bất cứ sự kiện nào làm thay đổi xung (từ thấp lên cao và ngược lại)
  - Tạo tốc độ Baud trong truyền thông nối tiếp

# Các thanh ghi dùng trong Timer/Counter


---

- Thanh ghi TH0, TL0
- Thanh ghi TH1, TL1
- Thanh ghi TMOD
- Thanh ghi TCON

# Các thanh ghi dùng trong Timer/Counter

---

- Thanh ghi Timer 0 và Timer 1



# Các thanh ghi dùng trong Timer/Counter

---

## ▪ Thanh ghi TMOD

GATE	C/T	M1	M0	GATE	C/T	M1	M0
Timer 1				Timer 0			

- **Gate** 0: Timer/Counter chỉ hoạt động khi bit TRx được set  
1: Timer/Counter chỉ hoạt động khi bit TRx được set và chân INTx được đặt ở mức cao
- **C/T** 0: Timer/Counter hoạt động như một bộ định thời  
1: Timer/Counter hoạt động như một bộ đếm

# Các thanh ghi dùng trong Timer/Counter

---

## ▪ Thanh ghi TMOD


M0	M1	Chế độ	Hoạt động
0	0	0	Chế độ timer 13 bit; 8 bit nạp cho thanh ghi THx và 5 bit nạp cho thanh ghi TLx
0	1	1	Chế độ timer 16 bit; 8 bit nạp cho thanh ghi THx và 8 bit nạp cho thanh ghi TLx
1	0	2	Chế độ 8 bit tự nạp lại ; THx sẽ chứa giá trị nạp, mỗi lần tràn giá trị nạp sẽ được chuyển sang TLx
1	1	3	Chế độ chia tách

# Các thanh ghi dùng trong Timer/Counter

---

## ■ Thanh ghi TMOD

- Ví dụ: tìm giá trị nạp cho TMOD, nếu muốn sử dụng timer 0 ở mode 2; sử dụng nguồn xung từ thạch anh của 8051 và sử dụng lệnh để khởi động/ dừng Timer





# Các thanh ghi dùng trong Timer/Counter


---

## ■ Thanh ghi TMOD


- Ví dụ:

- MOV TMOD,#00000001B ; Timer 0 mode 1, hoạt động như bộ định thời
- MOV TMOD,#20H ; Timer 1 mode 2, hoạt động như bộ định thời
- MOV TMOD,#12H ; Timer 1 mode 1, Timer 0 mode 2,
- MOV TMOD,#00000101B ; Timer 0 mode 1, hoạt động như bộ đếm
- MOV TMOD,#00001001B ; Timer 0 mode 1, dùng nguồn xung từ bên ngoài (GATE=1)


# 4 chế độ hoạt động Timer/Counter




Mode 0 : 13-bit counter  
(4048 mode)



Mode 1 : 16-bit counter



Mode 2 : 8-bit auto reload counter



Mode 3 : two 8-bit counter  
the other counter will not  
output overflow (interrupt)

# Các thanh ghi dùng trong Timer/Counter

- Thanh ghi TCON (thanh ghi điều khiển Timer)

(MSB)				(LSB)			
TF1	TR1	TF0	TR0	IE1	IT1	IE0	IT0
Timer 1		Timer0		for Interrupt			

- 4 bit thấp dùng điều khiển **Ngắt**
  - **IT0**
  - **IE0**
  - **IT1**
  - **IE1**
- 4 bit cao dùng điều khiển cho **Timer**
  - **TR0: Bit khởi động/ dừng Timer 0**
  - **TR1: Bit khởi động/ dừng Timer 1**
    - **TRx=0: STOP**
    - **TRx=1: START**

# Các thanh ghi dùng trong Timer/Counter

- Thanh ghi TCON (thanh ghi điều khiển Timer)

(MSB)				(LSB)			
TF1	TR1	TF0	TR0	IE1	IT1	IE0	IT0
Timer 1		Timer0		for Interrupt			

- 4 bit cao dùng điều khiển cho **Timer**
  - **TF0: Bit báo tràn cho Timer 0**
  - **TF1: Bit báo tràn cho Timer 1**
    - **TRx=0: STOP**
    - **TRx=1: START**
  - **Khởi tạo TFX=0, khi Timer đếm từ FFFFh về 0000h, cò tràn TFX=1**

# Ví dụ dùng Timer/Counter để tạo trễ

---

- Ví dụ 1: dùng Timer 1 ở mode 1 để tạo khoảng thời gian trễ là  $100\mu\text{s}$ , biết  $f_{\text{xtal}}=12\text{MHz}$ 
  - $f_{\text{xtal}}=12\text{MHz} \rightarrow f_{\text{timer}}=12\text{MHz}/12=1\text{MHz}$  &  
 $t_{\text{timer}}=1/1\text{MHz}$
  - $t_{\text{delay}}=100\mu\text{s}$ ,
  - Giá trị nạp cho TH1 và TL1:  $65536-100=65436=\text{FF9Ch}$   
 $\rightarrow \text{TH1}=\text{FFh}$  và  $\text{TL1}=\text{9Ch}$

# Ví dụ dùng Timer/Counter để tạo trễ

---

- Ví dụ 1: dùng Timer 1 ở mode 1 để tạo khoảng thời gian trễ là  $100\mu\text{s}$ , biết  $f_{\text{xtal}}=12\text{MHz}$

- Viết chương trình tạo trễ  $100\mu\text{s}$

```
MOV TMOD, #00010000B; Timer 1, mode1
MOV TL1, #9CH ; Initial count
MOV TH1, #0FFH ; -100 = FF9CH
SETB TR1 ; start Timer 1
WAIT: JNB TF1, WAIT ; wait for overflow
CLR TF1 ; clear overflow flag
CLR TR1 ; stop Timer 1
```

## **Chú ý: một số cách viết khác có thể sử dụng**

```
MOV TL1, #9CH ⇔ MOV TL1, #LOW(-100)
MOV TH1, 0FFH ⇔ MOV TH1, #HIGH(-100)
WAIT: JNB TF1, WAIT ⇔ JNB TF1, $
```

# Ví dụ dùng Timer/Counter để tạo trễ

- Ví dụ 1: dùng Timer 1 ở mode 1 để tạo khoảng thời gian trễ là 5s, biết  $f_{\text{xtal}}=12\text{MHz}$ 
  - Biết  $1\text{s}=50\text{ms}\cdot 20$
  - Giá trị nạp cho TH1 và TL1 là;  $65.536-50.000=15536=3\text{CB}0\text{h}$   
→ TH1=3Ch & TL1=B0h
  - CODE

```
MOV TMOD, #00010000B ; Timer 1, mode1
```

```
...
```

```
DELAY1S:
```

```
MOV R7, #20 ; 20 loops
```

```
LOOP: MOV TL1, #0B0h ;
```

```
MOV TH1, #3Ch ;
```

```
SETB TR1 ; start Timer 1
```

```
JNB TF1, $ ; wait for overflow
```

```
CLR TF1 ; clear overflow flag
```


```
CLR TR1 ; stop Timer 1
```

```
DJNZ R7, LOOP
```

```
RET
```

# Ví dụ dùng Timer/Counter để tạo trễ

- Ví dụ 1: dùng Timer 0 ở mode 1 để tạo xung vuông có tần số 10Hz trên chân, biết  $f_{\text{xtal}} = 12\text{MHz}$ 
  - $f = 10\text{Hz} \rightarrow T = 1/f = 100.000\mu\text{s}$
  - Vậy  $t_H = 50.000\mu\text{s}$  &  $t_L = 50.000\mu\text{s}$





## ▪ Code

```
MOV TMOD, #01H ;Timer 0,mode 1(16-bit timer mode)
LOOP: MOV TH0, #HIGH(-50000); high byte of -50,000
      MOV TL0, #LOW(-50000) ; low byte of -50,000
      SETB TR0 ; start timer
WAIT: JNB TF0, WAIT ; wait for overflow
      CLR TR0 ; stop timer
      CLR TF0 ; clear timer overflow flag
      CPL P1.0 ; toggle port bit
      SJMP LOOP ; repeat
```




# Ví dụ dùng Timer/Counter để tạo trễ

- Ví dụ 1: dùng Timer 0 ở mode 1 để tạo xung vuông có tần số 10Hz trên chân P1.0, biết  $f_{\text{xtal}}=12\text{MHz}$ 
  - $f=10\text{Hz} \rightarrow T=1/f=100.000\mu\text{s}$
  - Vậy  $t_H=50.000\mu\text{s}$  &  $t_L=50.000\mu\text{s}$
- Sử dụng Oscilloscope để quan sát



# Ví dụ dùng Timer/Counter để tạo trễ

- Viết chương trình để đếm số lần nhấn công tắc ở chân P3.4 và hiển thị trên 8 led đơn được kết nối ở chân P1.



# Ví dụ dùng Timer/Counter để tạo trễ

---

- Viết chương trình để đếm số lần nhấn công tắc ở chân P3.4 và hiển thị trên 8 led đơn được kết nối ở chân P1.

- **CODE**

```
ORG 0000H
```

```
    MAIN:
```

```
    MOV TMOD,#00000101B;Timer 0, 16 bit, external  
clock
```

```
    ; (counter operation)
```

```
    ; Gate=0, C/T=1, M1 M0 = 01
```

```
    MOV TH0,#0
```

```
    MOV TL0,#0
```

```
    SETB TR0 ;Start Timer
```

```
LOOP: MOV A,TL0 ;Read Timer
```


```
    MOV P1,A ;Display on Bar-LED
```

```
SJMP LOOP
```

```
END
```

# Ví dụ dùng Timer/Counter để tạo trễ

- Viết chương trình để đếm số lần nhấn công tắc ở chân P3.4 và hiển thị trên Led 7 thanh đơn được kết nối ở chân P1.



# Ví dụ dùng Timer/Counter để tạo trễ

---

- Viết chương trình để đếm số lần nhấn công tắc ở chân P3.4 và hiển thị trên Led 7 thanh đơn được kết nối ở chân P1.
- CODE

```
ORG 0000H
MAIN: MOV TMOD,#00000101B
MOV TH0,#0
MOV TL0,#0
SETB TR0
LOOP: MOV A,TL0
CJNE A,#10,NEXT
CLR A
MOV TL0,#0
NEXT: ACALL DISPLAY
SJMP LOOP
DISPLAY:
ACALL BCDTO7SEG
MOV P1,A
RET
```

```
BCDTO7SEG:
MOV DPTR,#MALED
MOVC A,@A+DPTR
RET
MALED: DB 40h,79h,24h,30h,19h
DB 12h,02h,78h,00h,10h
DONE: NOP
END
```

# Timer/Counter

---